NASA TECHNICAL NOTE

NASA TN D-6526

C.1

NASA TN D-6526

SOME LIMITATIONS ON THE USE OF DAMPING IN SHORT PRESSURE PROBES

by Ted W. Nyland and Robert C. Anderson

Lewis Research Center
Cleveland, Ohio 44135

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • OCTOBER 1971

1. Report No.
NASA TN D-6526

2. Government Accession No.

3. Recipient's Catalog No.

4. Title and Subtitle
SOME LIMITATIONS ON THE USE OF DAMPING IN SHORT PRESSURE PROBES

5. Report Date
October 1971

6. Performing Organization Code

7. Author(s)
Ted W. Nyland and Robert C. Anderson

8. Performing Organization Report No.
E-6377

9. Performing Organization Name and Address
Lewis Research Center
National Aeronautics and Space Administration
Cleveland, Ohio 44135

10. Work Unit No.
126-61

11. Contract or Grant No.

12. Sponsoring Agency Name and Address
National Aeronautics and Space Administration
Washington, D. C. 20546

13. Type of Report and Period Covered
Technical Note

14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

The analytical study described in this report examines changes in the useful frequency response of short pressure probes resulting from variations in the viscous damping within the probe tube. The study is based on the use of the Bergh and Tijdeman recursion formula, which was modified to allow the addition of arbitrary amounts of viscous loss in any tube section. Results show that the useful response range can be increased with damping but is dependent on the operating temperature and pressure levels. At 300 K, only a 15-percent variation in average pressure is allowable if the useful response is to be a factor of three higher than that of an undamped probe. This variation increases to 20 or 30 percent at temperatures above 600 K. A computer program based on the modified recursion formula is given in an appendix.

17. Key Words (Suggested by Author(s))
Pressure measurements
Pressure probe response
Damping
Frequency response
Amplitude ratio
Phase angle

18. Distribution Statement
Unclassified - unlimited

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	26	$3.00

* For sale by the National Technical Information Service, Springfield, Virginia 22151

SOME LIMITATIONS ON THE USE OF DAMPING IN SHORT PRESSURE PROBES

by Ted W. Nyland and Robert C. Anderson

Lewis Research Center

SUMMARY

An analytical study is described in which variations in the useful frequency response of a short pressure probe are calculated as a function of viscous losses (damping) and operating conditions. The purpose of the damping is to increase the useful frequency response of the probe where useful response is defined as that frequency range over which the amplitude ratio is 1.0±0.05. The study is based on the use of the Bergh and Tijdeman recursion formula, which was modified to allow for the inclusion of sections with sets of parallel tubes. A computer program used to calculate the response of probes with the modified recursion formula is given in the appendix. Results indicate that damping can be used to increase the useful response of a probe by a factor of three or more. The gain in useful response, however, is highly dependent on the operating pressure and temperature level. For a temperature of 300 K and average pressure of 10 newtons per square centimeter, a factor of three in useful response is achieved only over a 15-percent variation in the average pressure. At 300 K, the size of capillary tubes required to get useful damping is strongly dependent on pressure level. Tube sizes used in this study ranged from an 0.013-centimeter inside diameter for use at 11 newtons per square centimeter to 0.0088-centimeter inside diameter at 26 newtons per square centimeter.

INTRODUCTION

A number of experiments being conducted at the Lewis Research Center have required the measurement of transient pressures. Typical of these experiments is one in which the performance and stall limits of a turbofan engine (ref. 1) were determined. In this experiment, transient pressures were measured with probes consisting of short lengths of small-diameter tubing connected to a housing that held a miniature pressure transducer, resulting in a lightly damped measuring system. All of the probes in this experiment were of a minimum length as determined by the physical constraints imposed by the test environment and were designed for maximizing the natural frequency. For this experiment, it had been estimated that a useful response of 500 hertz was required, where useful response is arbitrarily chosen to be that frequency at which the amplitude ratio is 1.00 within ±5 percent. In other test programs being conducted at Lewis, useful frequency responses up to 1500 hertz are desired. To extend the useful response of many of these probes, a potentially attractive method involves increasing the viscous damping in the probe.

Generally speaking, most transient pressure probes can be considered to be lightly damped, second-order systems. Typical probes have undamped natural frequencies in the 2000- to 3000-hertz range with useful frequencies of 20 percent of the natural frequency (see ref. 2 for second-order response equation). With optimum damping, the useful frequency of a second-order system can be increased to better than 80 percent of the natural frequency. This more than quadruples the useful response range over that of an undamped probe and thus offers an attractive solution to the problem. It should be pointed out that the increase in useful response is not achieved without some sacrifice. The sacrifice in this instance is that a phase shift is introduced into the measurement which may or may not be acceptable in given measurement situations.

Experimental data have shown that the simple second-order system equations do not accurately predict the amplitude response over the entire frequency range of probes of complex geometry and/or those with appreciable damping. However, Iberall (ref. 3) developed an equation, not based on a second-order system approach, from which the amplitude response of a single tube and volume pressure transmission system can accurately be calculated. Iberall's equation is derived from the fundamental flow equations (that is, the Navier-Stokes equation, the equation of continuity, the equation of state, and the energy equation) and assumes that the sinusoidal disturbances are very small compared with the static conditions in the system. Goldschmied (ref. 4) has shown that available data correlate well with Iberall's equation. Following Iberall's approach, Bergh and Tijdeman (ref. 5) have developed a recursion formula from which the response of a series connected set of tube and volume systems can be computed. The Bergh and Tijdeman formula reduces to the Iberall equation for the case of a single tube and volume system. References 5 and 6 indicate that good agreement is achieved between experimental data and the recursion formula for many types of tube systems. As the recursion formula appears to adequately predict the response of many types of pressure measuring systems, it, along with a modified form discussed later, was used for all computations described herein.

An analytical study was made at Lewis for the purpose of determining what gains could be made in the useful response by increasing the damping in short pressure probes and examining some of the problems that might arise in practice. It was also of interest in the study to determine what changes would occur in the useful frequency response when operating at off-design temperature or pressure conditions in a probe. This report is a summary of that study. A discussion of some of the practical problems associated with the use of various damping materials is also presented.

ANALYTICAL MODELS AND RESPONSE EQUATIONS

A diagram of the basic probe used in the study is shown in figure 1. This probe consists of a 2.54-centimeter long section of tubing connected to a small cavity containing a miniature pressure transducer. The inside tube diameter is 0.325 centimeter.



Figure 1. - Diagram of basic probe. (Dimensions are in centimeters.)

The volume ratio of the probe, defined as the ratio of cavity volume to tube volume, is 0.027 and is considered representative of a minimum volume ratio that could be achieved in an actual probe design. The amplitude and phase angle response of this probe were computed from the recursion formulas of reference 5. The subroutines used in these computations are listed in reference 6.

The useful response of the basic probe is shown in figure 2. Plotted are curves of constant useful frequency as a function of pressure and temperature. As an example, at 300 K and 10 newtons per square centimeter, the useful response is seen to be 625 hertz. Any values of useful response discussed in this report can be referenced to the values in this figure since these represent the response of an undamped probe.

Two approaches could conceivably be used to increase the damping in the basic probe. In the first approach, one might consider simply working with smaller diameter probes and determining their effect on the useful response. The second approach would involve filling the probe tube with an insert of some material that would increase the viscous losses while maintaining a large cross sectional area for flow. The insert might take the form of a porous type of material such as compressed steel wool or it might take the form of a set of small-diameter parallel tubes. In both forms, the flow area would be made up of a large number of small-diameter passages that would result in



Figure 2. - Lines of constant useful frequency response as a function of temperature and pressure of an undamped probe.



Insert designation	Inside diameter	Number of tubes
I-204	0.02276	204
I-639	.01286	639
I-1367	.00879	1367

Figure 3. - Basic probe with parallel-tube insert section. (Dimensions are in centimeters.)

high viscous energy losses. For this report, three parallel-tube inserts were examined; their diameters, lengths, and numbers are given in figure 3. These inserts replace the 2.54-centimeter-long by 0.325-centimeter-diameter tube in the basic probe.

For computational purposes, the flow characteristics of the parallel-tube insert can be introduced into the recursion equations of reference 5 as follows. The general system, consisting of N tube and volume sections, used in the derivation of the recursion



Figure 4. - Tube-volume system of Bergh and Tijdeman derivation.

formula is shown in figure 4. In addition to the variables used in the derivation, we define a variable C_j, which is equal to the number of parallel tubes associated with the j^{th} section of the system. Likewise, there is a variable C_{j+1} for the $j+1$ section. These variables are included in the derivation by simply multiplying equations (42) and (45) of reference 5 by C_j and C_{j+1}, respectively. Equation (42) represents the total mass flow leaving the tube set in the j^{th} section; equation (45) represents the mass flow entering the tube set in the $j+1$ section. These variables are then carried through the remainder of the derivation, producing a modified form of the recursion formula:

$$\frac{P_j}{P_{j-1}} = \left[\cosh(\varphi_j L_j) + \frac{V_{v_j}}{C_j V_{t_j}}(1 + \sigma_j)n_j \varphi_j L_j \sinh(\varphi_j L_j)\right.$$

$$\left. + \frac{C_{j+1} D_{j+1}^2 \varphi_{j+1} J_2(\alpha_{j+1}) J_o(\alpha_j) \sinh(\varphi_j L_j)}{C_j D_j^2 \varphi_j J_o(\alpha_{j+1}) J_2(\alpha_j) \sinh(\varphi_{j+1} L_{j+1})}\left(\cosh(\varphi_{j+1} L_{j+1}) - \frac{P_{j+1}}{P_j}\right)\right]^{-1}$$

where

$$\sigma_j = \frac{P_s}{V_{v_j}}\left(\frac{\Delta V}{\Delta P}\right)$$

$$\varphi_j = \frac{\omega}{\sqrt{\dfrac{\gamma P_s}{\rho_{s_j}}}} \sqrt{\frac{J_o(\alpha_j)}{J_2(\alpha_j)}} \sqrt{\frac{\gamma}{n_j}}$$

$$\alpha_j = i^{3/2}\sqrt{\frac{\omega D_j^2 \rho_{s_j}}{4\mu_j}}$$

$$n_j = \frac{1}{1 + \frac{\gamma - 1}{\gamma}\,\frac{J_2\left(\alpha_j\sqrt{Pr}\right)}{J_0\left(\alpha_j\sqrt{Pr}\right)}}$$

(The symbols are defined in the appendix.) A set of FORTRAN IV subroutines based on the preceding modified version of the recursion formula is given in the appendix.

For all computations, except those discussed in the section Analysis of a Typical Damped Probe Application, N was equal to 1. In computing the frequency response of a given probe, the density of air at various temperatures and pressures was calculated from the perfect gas law equation. The air viscosity was calculated from an equation given in table 1-b of reference 7 (p. 10). A summary of these equations and other property values used in the recursion formulas are given in table I.

TABLE I. - PROPERTY VALUES AND EQUATIONS

USED IN COMPUTATIONS

Temperature, K	T
Pressure, N/cm^2	P
Specific heat ratio	1.4
Prandtl number	0.7
Density, g/cm^3	3.485×10^{-2} P/T
Viscosity[a], g/(sec)(cm)	$1.458\times10^{-5}\ T^{1.5}/(T + 110.4)$

[a]From ref. 7.

EFFECTS OF TUBE DIAMETER ON THE USEFUL FREQUENCY OF PROBES

On first thought, it would seem reasonable that the useful frequency response range of a transient pressure probe could be increased by using a smaller diameter tube in the probe. The small tube would result in an increase in viscous energy losses and therefore an increase in the effective damping of the probe. However, to gain in useful response, one must maintain a high natural frequency. The natural frequency of a probe is dependent on volume ratio and probe length.

As indicated in reference 6, to maintain a high natural frequency for a given length probe requires that the cavity volume be reduced proportionately with reductions in the square of the tube diameter. If the volume ratio is held constant, an increase in the useful response range can be achieved using smaller diameter tubes. This is illustrated in figure 5, which shows curves of the calculated useful frequency as a function of tube diameter for constant volume ratio and for constant transducer cavity volume. As is shown by the curve for constant volume ratio, the useful frequency of the basic probe could be optimized by using tube diameters of about 0.012 centimeter while maintaining a volume ratio of 0.027. However, it should be pointed out that this curve is representative of an ideal probe design, as one cannot build such a probe when using available miniature pressure transducers and allowing reasonable machining and construction tolerances.

A more realistic design condition is represented by the second curve in figure 5. This curve shows the computed useful frequency for the basic probe assuming that the volume is held constant while the tube diameter is varied. It is seen that the optimum response is achieved when using as large a diameter tube as possible and that no increase in the useful frequency can be obtained when using small diameter tubes.



Figure 5. - Useful frequency as function of tube diameter. Temperature, 295 K; pressure, 9.92 newtons per square centimeter.

FREQUENCY RESPONSE CALCULATIONS FOR PROBES WITH PARALLEL-TUBE INSERTS

Typical response curves calculated with the modified recursion formulas for a probe with a parallel-tube insert are shown in figure 6. Plotted are the amplitude and phase angle response curves for the basic probe and for the basic probe with the I-204 and I-639 inserts. The total cross sectional area for flow of the basic probe with and without the inserts are equal. The experimental data points are for the basic probe taken from reference 6 and show good agreement with the computed curve. As would be expected, increasing the number of tubes while decreasing their diameter results in a lowering of the peak amplitude ratio. In general, it is seen that, by use of different inserts, the amplitude response can be made to vary in a manner similar to that which occurs when the damping coefficient in a second-order system is changed. Both a reduction in the amplitude ratio and the introduction of a phase shift occurs. For the basic probe and the I-204 insert, the intersection of the amplitude curves with the ±5 percent (±0.42 dB) lines show useful frequency responses to about 600 hertz. By using the I-639 insert, the useful response has been extended to about 1300 hertz and results in more than doubling the



Figure 6. - Frequency response curves for basic probe and probes with I-204 and I-639 inserts at temperature of 300 K and pressure of 10 newtons per square centimeter.

response range of the basic probe. Thus, it appears that the useful response of a probe can be increased by use of these inserts.

TEMPERATURE AND PRESSURE EFFECTS ON THE RESPONSE OF A DAMPED PROBE

The amount of damping achieved with a given insert is a function of the density and viscosity of the fluid in the probe. Since these properties depend on temperature and pressure, the useful response of the probe also is temperature and pressure dependent. Figure 7 shows some typical response curves for a probe with an I-639 insert. In figure 7(a), response curves for the probe operating at 250, 300, and 350 K are shown for an assumed constant pressure of 10 newtons per square centimeter. The useful response is seen to change from 1000 to 1275 to 750 hertz at these temperatures. In figure 7(b), response curves for pressures of 5, 10, and 15 newtons per square centimeters are shown for an assumed constant temperature of 300 K. For these pressures, the useful response varies from 300 to 1275 to 750 hertz. The useful response for the basic probe is around 650 hertz. Thus, there exists a range of limited temperatures and pressures over which an insert such as an I-639 could be used for extending the useful frequency range of a transient pressure measuring probe.

A more complete picture of the range of temperatures and pressures over which a probe with an I-639 insert could find use is shown in figure 8. Plotted are lines of constant useful frequency as a function of temperature and pressure. As is shown in the figure, there is a region in the temperature and pressure domain in which the probe's useful response is greatly increased. As the operating temperature and pressures increase, the range in temperature and pressure over which this probe could be used also increases. However, at temperatures such as 300 K, the probe is limited to operating at 10 ± 1.5 newtons per square centimeter while having a useful response of 1800 hertz or more. For general probe applications, this small allowable variation in operating pressure imposes a severe limitation on the usefulness of a damped probe.

Figure 9 is a similar plot of lines of useful frequency as a function of pressure and temperature for a probe with an I-1367 insert. The overall shapes of the lines for this probe are the same as those shown in figure 8 but are more broadly spaced and displaced to the right along the pressure axis. At 300 K, this probe could find use at 25 ± 3.0 newtons per square centimeter and still maintain a useful response to at least 1800 hertz. Again, this is a fairly small allowable operating pressure range and severely restricts the use of damped probes of this type when operating at low temperatures. Even with a relaxation of the useful response to 1200 hertz (approximately doubling the range of the basic probe), the pressure range for operation would be increased



(a) Showing effects of temperature.



(b) Showing effects of pressure.

Figure 7. - Frequency response curve for a probe with an I-639 insert.



Figure 8. - Lines of constant useful frequency as function of pressure and temperature for a probe with an I-369 insert.



Figure 9. - Lines of constant useful frequency as function of pressure and temperature for a probe with an I-1367 insert.

only to ±2 newtons per square centimeter for the I-639 insert and to ±6 newtons per square centimeter for the I-1367 insert. In general then, for measurements around 300 K, it does not appear practical to use damping to increase the useful response of probes unless the average pressure is known and tightly held to within 10 or 20 percent of the probe design value. At higher temperatures, this restriction in operating conditions becomes less severe and might allow one to use damping for increased frequency response. We feel that any type of materials used to increase the damping in a probe will have similar operating restrictions and that, in general, these results presented for the parallel-tube inserts are applicable to a wide variety of damping materials.

ANALYSIS OF A TYPICAL DAMPED PROBE APPLICATION

In turbojet engine experiments, transient pressure probes are sometimes positioned between compressor stages in regions of elevated temperature. These temperatures are generally higher than the allowable operating temperature for the miniature pressure transducer, and water cooling is normally used to maintain a temperature that is compatible with the transducer's operating requirements. This temperature is normally around 300 K. The temperature at the probe inlet may range from 300 K to more than 700 K, depending on the probe's position in the compressor. The useful response then becomes a function of the temperature distribution between the inlet and cavity volume of the probe. One, therefore, must take this temperature distribution into consideration when determining the useful response range of a probe.

A set of calculations were made to determine what useful response could typically be achieved under test condition which resulted in a thermal gradient between the inlet and cavity volume of a damped probe. A linear temperature distribution was assumed to exist over the probe length with the cavity volume at 300 K and the probe inlet at a temperature corresponding to the pressure levels as shown in figure 10. The curve in figure 10 corresponds approximately to the temperature and pressure distribution that would occur in a compressor with an inlet temperature of 300 K, an inlet pressure of 5 newtons per square centimeter, and an efficiency η of 88 percent. The results of the response calculations are shown in figure 11 where it was assumed, for computational purposes, that the probe consisted of 10 equal length sections (N = 10). Plotted are the ratios of useful response for probes with the I-639 and I-1367 inserts to the useful response of a basic probe. From the curves in figure 11, the useful response of a probe operating at average pressures of 20±3 newtons per square centimeter could be doubled by use of an I-639 insert. Similarly, the use of an I-1367 insert would double the useful response range of a probe designed for average pressures of 60±15 newtons per square centimeter. These operating levels are somewhat higher than those that would exist if the probes were held at constant temperature (see figs. 8 and 9). This points out the necessity of knowing the temperature distribution in a probe when one is considering the application of damping to short probes. As was the case for the constant-temperature probe at low pressures, the allowable operating range is very small and may limit the general application of the inserts. At high pressures, one could consider the use of in-



Figure 10. - Probe inlet temperature as function of average pressure level. Compressor inlet temperature T_C, 300 K; compressor inlet pressure P_C, 5 newtons per square centimeter; compressor efficiency η, 0.88; specific heat ratio γ, 1.4.



Figure 11. - Example of change in useful response as function of pressure level for two damped probes assuming a linear temperature gradient between probe inlet and cavity volume. Probe inlet and cavity volume temperatures are given in figure 10. The response ratio is defined as the ratio of useful frequency for a damped probe to that of the basic probe.

serts to increase the useful response by at least a factor of two over that obtained with a probe with no insert. It is interesting to note that by use of an I-639 insert, the useful response would be equal to or higher than that achieved with an undamped probe at all pressures above 12 newtons per square centimeter over the range indicated. The same is true for an I-1367 insert above 35 newtons per square centimeter.

INSERT DESIGN CONSIDERATIONS

Preliminary screening of materials that may be suitable for use as inserts can be made by considering two basic design requirements. The first requirement is that the probe and insert combination should have a small volume ratio. For a damped probe, the volume ratio is defined as the ratio of cavity volume to the total volume of the flow passages in the insert. The reason for this requirement is illustrated in figure 12.



Figure 12. - Useful frequency as function of number of 0.01286-centimeter-inside-diameter tubes at 330 K and 10 newtons per square centimeters.

Shown is a plot of the useful response of a damped probe as a function of the number of 0.0128-centimeter-diameter tubes in the insert. Also indicated along the curve are corresponding values of volume ratio. (For these calculations, the volume ratio is inversely proportional to the number of tubes.) Initially the curve rises sharply and becomes asymptotic to some value of frequency at which large increases in the number of tubes (decreases in the volume ratio) result in only small increases in the useful response. Depending on how much reduction in the optimum useful frequency response is acceptable, one can place lower limits on the number of tubes that would be required in the case of a parallel-tube insert, or upper limits on the volume ratio of the probes with porous material inserts. For most damped probe designs, it is recommended that one consider only materials that, when used in a probe, would result in volume ratios in the 0.05 range. This 0.05 volume ratio would result in a decrease in response of not more than about 15 percent compared with a system with zero volume ratio.

The second design requirement is related to the fact that the individual passageways making up the total flow area be small enough to obtain adequate viscous losses in the system. For a parallel-tube insert, the required tube diameters fall into the 0.01-centimeter-diameter range, their actual size being dependent on the operating temperature and pressure. As shown in figures 8 and 9 a design temperature of 300 K and a pressure of 11 newtons per square centimeter would require tubes of 0.0128-centimeter inside diameter. Changing the design pressure to 26 newtons per square centimeter would require tubes of 0.00879 centimeter diameters. These tubes are small enough so that manufacturing tolerances become a problem.

Other than an actual parallel-tube set, inserts could be made from sintered metal-fiber materials. The actual flow passages in these materials are irregular, consisting of spaces between the metal fibers. The hole sizes in these materials would generally have to be smaller than those associated with a parallel-tube insert in order to achieve similar damping effects. So, in general, one can consider using materials that have large cross sectional areas for flow and with hole sizes in the range of 0.01 centimeter for a tube system and less than that value for a fiber metal type material.

From an analytical standpoint, it is very difficult and beyond the scope of this report to introduce the damping characteristics of porous materials into the recursion formula. Any flow parameters specified by the manufacturer of porous materials are normally measured under steady-state flow conditions. These parameters cannot readily be converted to useful parameters for the case of an oscillatory flow. One, therefore, would probably have to measure the frequency response of the probe with a particular insert. Then by trial and error, the modified recursion equation could be used to determine an equivalent parallel-tube insert whose frequency response curve matched the experimental data points. Then, trends in the response characteristics for other temperature and pressure conditions could be obtained through calculations based on the parallel-tube equivalent.

CONCLUDING REMARKS

Ideally, the useful frequency response range of a transient pressure measuring probe can be increased by introducing damping into the probe body. For a given size probe, it has been shown that just decreasing the probe tube diameter will not result in an increase in the useful response. At small tube diameters, the volume ratios becomes the controlling parameter and, therefore, the cavity volume must be decreased proportionally with the change in tube diameter to achieve a gain in the useful response. Inserts consisting of sets of parallel tubes can be used to increase the viscous losses in probes. For measurements at standard conditions, diameters for these parallel tubes would be in the 0.01-centimeter range. The useful response of probes with inserts becomes strongly dependent on the operating temperature and pressure of the gas in the probe. At 300 K and 10 newtons per square centimeter, one can raise the useful response by a factor of three if the operating pressure is held within 15 percent of the design pressure. At higher temperature and pressure levels, this operating restriction becomes somewhat less severe. For probes with inserts that are not of the parallel-tube type, the frequency response will normally have to be determined experimentally. Trends in the useful response curve can then be predicted by use of the recursion formula. When damping is introduced into probes, one must remember that a phase shift is introduced into the measurement that may or may not be acceptable in all measurement situations.

Lewis Research Center,
National Aeronautics and Space Administration,
Cleveland, Ohio, July 28, 1971,
126-61.

APPENDIX - COMPUTER SUBROUTINES FOR CALCULATING THE RESPONSE AT A FREQUENCY OF A MULTIPLE TUBE-VOLUME SYSTEM INCLUDING PARALLEL TUBE SECTIONS

A set of subroutines for computing the amplitude ratio and phase angle at a frequency for a multiple tube-volume system with parallel tube sections is described herein. The subroutines are based on the modified Bergh and Tijdeman equation as discussed in a previous section of this report. The symbols used in the modified equation and computer program are defined as follows:

Engineering symbol	Computer symbol	Definition
---	A	amplitude ratio
C	CONGL	number of tubes in given section
D	D	tube diameter
f	FREQ	frequency
i	-----	$\sqrt{-1}$
J_n	-----	Bessel function of first kind of order n
L	XL	tube length
N	N	number of tube-volume sections
P	---	amplitude of pressure disturbance
Pr	P	Prandtl number (usually equal to 0.7)
---	PA	phase angle
P_s	AMPRES	mean pressure
V_t	--------	volume of a single tube in a given section
V_v	V	cavity volume of given section
γ	GAMMA	specific heat ratio
μ	VIS	fluid dynamic viscosity
ρ_s	RHO	mean mass density
$\left(\frac{\Delta V}{\Delta P}\right)$	DVDP	change in volume V due to change in pressure

$\omega=2\pi f$	---	angular frequency
j	I	tube-volume section under consideration

The FORTRAN IV subroutines are referenced by the following call statement:

```
CALL AMPPHS (FREQ, A, PA, I, N)
```

where FREQ is the oscillatory frequency for which the amplitude ratio and phase angle are to be computed, N is the total number of series connected tube-volume sections, and I is the section for which the amplitude ratio and phase angle are being computed. FREQ, N, and I must be defined before calling the subroutine. The computed amplitude ratio P(I)/P(0) will be placed in A, and the computed phase angle in PA on completion of a computation.

The program accepts as many as 10 series-connected tube-volume sections in a system. The parameters for each section must be put into the subroutines through three common blocks referenced as follows:

```
COMMON/GEOMTY/D(10), XL(10), V (10), DVDP (10)

COMMON/THERMO/GAMMA, P (10), RHO (10), VIS (10), AMPRES

COMMON/CON/CONGL (10)
```

The geometry parameters, diameter (D), length (XL), volume (V), and volumetric displacement (DVDP), for each section are entered through the GEOMTY common block. Zero values (0.0) may be entered for any or all values of V and DVDP. The thermodynamic fluid properties, ratio of specific heat (GAMMA), density (RHO), viscosity (VIS), and ambient pressure (AMPRES), are entered through the THERMO common block. The ratio of specific heats and ambient pressure are assumed to be equal in all sections. The number of parallel tubes (CONGL) in each section is entered through the CON common block. For a single tube section, a 1.0 must be specified for the CONGL value associated with that section.

The program used for generating the value of the Bessel function is limited in use to Bessel function of integer order with arguments of the form C (1 - i) where C is a real constant. Note: In computing the value for a Bessel function, very small numbers are generated that may result in machine underflows. It is suggested that use be made of a major underflow option if available on the machine being used. These underflows have been found to occur in the evaluation of statements 1000 and 2000 in the SERIES subroutine.

A listing of the nine subroutines required for computing the amplitude ratio and phase angle at one frequency for a series connected system of tube-volume sections including sections with parallel tube sets using the modified Bergh and Tijdeman equation is given next.

```
$IBFTC SUBAMP  LIST
      SUBROUTINE AMPPHS(FREQ,A,PA,I,J)
C
C
C     THIS ROUTINE CALCULATES THE AMPLITUDE RATIO (A) P(I)/P(0)
C AND PHASE ANGLE (PA)  FOR A SERIES CONNECTION OF J TUBE-VOLUME
C SYSTEMS GIVEN FREQUENCY  'FREQ'.
C     THE SYSTEM GEOMETRY IS ENTERED THROUGH COMMON BLOCK /GEOMTY/.
C     THE THERMODYNAMIC CONDITIONS ARE ENTERED THROUGH COMMON BLOCK /THERMO/.
C
C
      DIMENSION PRATIO(10)
      COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
      COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
      COMMON/XFREQ/FREQ1
      COMMON/CON/CONGL(10)
      COMPLEX  PRATIO,RECURS
      FREQ1 = FREQ
      RECURS = 1.0
      DO 10  M=1,10
10     PRATIO(M) = 0.0
      N = J
C
C CALCULATE SUCCESSIVE PRESSURE RATIOS
C
      DO 20  K=1,N
20    CALL RATIO(PRATIO,J+1-K,J)
C
C CALCULATE REQUIRED RATIO BY MULTIPLYING APPROPRIATE PRESSURE RATIOS
C
      DO30 K = 1,I
30    RECURS = RECURS*PRATIO(K)
      A= CABS(RECURS)
      PA= ATAN2(AIMAG(RECURS),REAL(RECURS))
C
C CONVERT RADIANS TO DEGREES
C
      PA= PA*57.2957795
      IF(PA.GT.0.0) PA = -360.+PA
      RETURN
      END
```

```
$IBFTC XRATIO  LIST
      SUBROUTINE RATIO(PRATIO,J,K)
C
C
C    THIS PROGRAM CALCULATES THE PRESSURE RATIO P(J)/P(J-1).
C
C
      DIMENSION PRATIO(10),VRATIO(10),CORRCT(10)
      COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
      COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
      COMMON/XFREQ/FREQ
      COMMON/CON/CONGL(10)
      COMPLEX PHI,JRAT,POLY,POLRAT,CCOSH,CSINH,PRATIO,TERM1,TERM2,TERM3,
     CTPHI1,TPHI2,JRATIO
      PI = 3.1415927
      VRATIO(J) = V(J)/(PI*D(J)**2*XL(J)/4.)
      VRATIO(J) = VRATIO(J)/CONGL(J)
      RFREQ = 2.*PI*FREQ
      IF(V(J).NE.0.0)GOTO1
      CORRCT(J) = 1.0
      GOTO3
1     CORRCT(J) = 1.0 + (DVDP(J)*AMPRES/V(J))
3     CONTINUE
      IF(J.EQ.K) GO TO 5
      CONRAT = CONGL(J+1)/CONGL(J)
      POLRAT = CSQRT(POLY(J)/POLY(J+1))
      RHORAT = SQRT(RHO(J+1)/RHO(J))
      VOLRAT = D(J+1)**2/(D(J)**2)
      TPHI2 = PHI(J+1)
      JRAT = CSQRT(
     JJRATIO(CSQRT(CMPLX(0.0,-(RFREQ*D(J+1)**2*RHO(J+1)/
     J( 4.0*VIS(J+1)))))))/
     JJRATIO(CSQRT(CMPLX(0.0,-(RFREQ*D(J  )**2*RHO(J  )/
     J( 4.0*VIS(J  )))))))
5     TPHI1 = PHI(J)
      TERM1 = CCOSH(TPHI1*XL(J))
      TERM2 = VRATIO(J)*CORRCT(J)*XL(J)*TPHI1*POLY(J)*CSINH(TPHI1*XL(J))
      IF(J.EQ.K) GO TO 2
      TERM3 = (CSINH(TPHI1*XL(J))/CSINH(TPHI2*XL(J+1)))*
     T        VOLRAT*RHORAT*JRAT*POLRAT*
     T        (CCOSH(TPHI2*XL(J+1))-PRATIO(J+1))
      TERM3 = TERM3*CONRAT
      GOTO6
2     TERM3 = CMPLX(0.0,0.0)
6     CONTINUE
      PRATIO(J) = 1.0/(TERM1+TERM2+TERM3)
      RETURN
      END
```

```
$IBFTC SERIE   LIST
      SUBROUTINE SERIES
C
C
C     THIS ROUTINE CALCULATES THE VALUE OF THE SERIES NEEDED FOR
C HANKELS ASYMPTOTIC EXPANSION
C
C    IN CALCULATING P(J) AND Q(J) NUMBERS WITH EXPONENTS LESS THAN -38
C MAY OCCUR.  THIS PROGRAM USES A BUILT-IN ROUTINE WHICH SETS P(J)
C AND/OR Q(J) EQUAL TO ZERO WHENEVER THESE SMALL NUMBERS OCCUR.
C
C
C
C
      COMPLEX Z,MU,PSUM,QSUM,FM,P(50),Q(50)
      COMMON/S1/X,Y,PSUM,QSUM
      COMMON /FMNAM/FM
      PSUM = (0.,0.)
      QSUM = (0.,0.)
      MU =4.*FM**2
      Z = CMPLX(X,Y)
      DO 100 J=1,50
      N = J-1
      FN =N
      IF(N .NE. 0) GO TO 30
20    P(J) =(1.,0.)
      Q(J) = (MU-1.)/(8.*Z)
      GO TO 90
30    FN2 = 2.*FN
      FN4 = 2.*FN2
      FN21 = FN2 + 1.
      FN41 = 2.*FN21
      IF(CABS(P(J-1)).EQ.0.) GO TO 50
 1000 P(J) = -((MU-(FN4-3.)**2)/(64.*(FN2-1.)*FN2))*((MU-(FN4-1.)**2)/Z
     $**2)*P(J-1)
      IF (J .GT. 2 .AND. CABS(P(J)) .GT. CABS(P(J-1)) ) P(J) = (0.,0.)
      GO TO 55
50    P(J) = (0.,0.)
55    IF(CABS(Q(J-1)).EQ.0.) GO TO 80
 2000 Q(J) = -((MU-(FN41-3.)**2)/(64.*FN2*FN21))*((MU-(FN41-1.)**2)/Z
     $**2)*Q(J-1)
      IF (J .GT. 2 .AND. CABS(Q(J)) .GT. CABS(Q(J-1)) ) Q(J) = (0.,0.)
      GO TO 90
80    Q(J) = (0.,0.)
90    PSUM = P(J)+PSUM
      QSUM = Q(J)+QSUM
      IF(CABS(P(J)).EQ.0. .AND. CABS(Q(J))  .EQ. 0.) GO TO 110
100   CONTINUE
110   CONTINUE
      RETURN
      END
```

```
$IBFTC BESSNN  LIST
      SUBROUTINE ZEEBES(J,N,Z)
C
C   THIS ROUTINE CALCULATES THE VALUE OF THE BESSEL FUNCTION OF THE
C FIRST KIND OF INTEGER ORDER. THE ARGUMENT IS OF THE FORM
C C(1 - SQRT(-1)) WHERE C IS A REAL CONSTANT LESS THAN 30 .
C
C                       J...BESSEL FUNCTION VALUE
C                       N....BESSEL FUNCTION ORDER
C                       Z...BESSEL FUNCTION ARGUMENT
C
C
      COMPLEX J,Z
      DOUBLE PRECISION DFN,T,R(2),C,DK,D,B
      D = REAL(Z)
      C = D * 1.4142135623730950D0
      CR = C
      DFN = FLOAT(N)
      IF(N.NE.2) GO TO 1
      T = 5.D-1
      GO TO 2
1     T = 1.D0
2     R(1) = T
      R(2) = 0.D0
      C = (C**2.) / 4.D0
      DK = 0.D0
      I = 1
      M = 0
5     DK = DK + 1.D0
      T = T * C / ((DK)*(DK+DFN))
      I = I + 1
      IF (T.LT.DABS(R(I)*1.D-8)) GO TO 7
      M = M + 1
      B = 1.D0
      IF(M.GT.1) B = -1.D0
      IF(M.EQ.3) M = -1
      R(I) = R(I) + B * T
      IF(I.EQ.2) I = 0
      GO TO 5
7     RR= R(1)
      RI = R(2)
      IF(N.NE.2) GO TO 8
      RI = (-1.) * R(1) * C
      RR =R(2) * C
8     J = CMPLX(RR,RI)
      RETURN
      END
```

```
$IBFTC XJRATO  LIST
      COMPLEX FUNCTION JRATIO(CARG)
C
C
C   THIS ROUTINE CALCULATES THE RATIO OF THE BESSEL FUNCTION OF THE
C FIRST KIND, ORDER 2 TO THE BESSEL FUNCTION OF THE FIRST KIND,
C ORDER 0 WITH COMPLEX ARGUMENT.
C
C
      COMPLEX FM,QSUM,PSUM,XPSUM(2),XQSUM(2),XJ(2),RATIO,XX,Y
     1Y,Z
      COMPLEX J2,J0,JRATIO,CARG
      COMMON /S1/X,Y,PSUM,QSUM
      COMMON /FMNAM/FM
      DATA PI/3.1415927/,M/-1/,Z/(0.0,1.0)/
      IF(REAL(CARG).GT.30.0) GO TO 40
C
C    IF THE REAL PART OF THE ARGUMENT IS GREATER THAN 30 USE IS MADE OF
C HANKELS ASYMPTOTIC EXPANSIONS (SEE NBS HANDBOOK OF MATHEMATICAL
C FUNCTIONS AMS NO. 55) TO CALCULATE THE BESSEL FUNCTION RATIO. THE P
C AND Q COEFFICIENTS USED IN THE EXPANSIONS ARE CALCULATED USING THE
C SERIES SUBROUTINE.
C
      GO TO 20
40    X = REAL(CARG)
      Y = -Y
      DO 2 J=1,2
      M = M + J
      XI = FLOAT(M)
      FM = CMPLX(XI,0.0)
      CALL SERIES
      XPSUM(J) = PSUM
2     XQSUM(J) = QSUM
      DO 3 K=1,2
      A = X + (0.75 - FLOAT(K))*PI
      XX = XPSUM(K) * COS(A)  - XQSUM(K) * SIN(A)
      YY = (XPSUM(K) * SIN(A) + XQSUM(K) * COS(A)) * Z
3     XJ(K) = XX + YY
       RATIO = XJ(2) / XJ(1)
      JRATIO = RATIO
      M = -1
      GO TO 30
20    CALL ZEEBES(J0,0,CARG)
      CALL ZEEBES(J2,2,CARG)
      JRATIO = J2/J0
30    CONTINUE
      RETURN
      END
```

```
$IBFTC SUBPHI  LIST
      COMPLEX FUNCTION PHI(J)
C
C        THIS PROGRAM CALCULATES A FUNCTION PHI WHICH IS AN
C     ATTENUATION PARAMETER.
C
      COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
      COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
      COMMON/XFREQ/FREQ
      COMPLEX POLY,PHI,JRATIO
      SN =6.2831854*FREQ*D(J)**2*RHO(J)/VIS(J)
      PHI=6.2831854*FREQ*SQRT(RHO(J)/(GAMMA*AMPRES))*
     P         CSQRT(1./JRATIO(CSQRT(CMPLX(0.0,-(SN/4.0)))))*
     P         CSQRT(GAMMA/POLY(J))
      RETURN
      END
```

```
$IBFTC SUBPLY  LIST
      COMPLEX FUNCTION POLY(J)
C
C        THIS ROUTINE CALCULATES A POLYTROPIC COEFFICIENT
C     (LABELED 'N' IN BERGH AND TIDJEMAN)
C
      COMMON/GEOMTY/D(10),XL(10),V(10),DVDP(10)
      COMMON/THERMO/GAMMA,P(10),RHO(10),VIS(10),AMPRES
      COMMON/XFREQ/FREQ
      COMPLEX JRATIO,CARG,POLY
      SN = 6.2831854*FREQ*D(J)**2*RHO(J)/VIS(J)
      ARG = P(J)*SN/4.0
      CARG= CSQRT(CMPLX(0.0,-ARG))
      POLY=1.0/(1.0+((GAMMA-1.0)/GAMMA)*JRATIO(CARG))
      RETURN
      END
```

```
$IBFTC FCSINH  LIST
      COMPLEX FUNCTION CSINH(ARG)
      COMPLEX ARG,CSINH
      CSINH= (CEXP(ARG)-CEXP(-ARG))/2.0
      RETURN
      END
```

```
$IBFTC FCCOSH  LIST
      COMPLEX FUNCTION CCOSH(ARG)
      COMPLEX ARG,CCOSH
      CCOSH = (CEXP(ARG)+CEXP(-ARG))/2.0
      RETURN
      END
```

REFERENCES

1. Braithwaite, Willis M.; and Vollmar, William R.: Performance and Stall Limits of a YTF30-P-1 Turbofan Engine with Uniform Inlet Flow. NASA TM X-1803, 1969.
2. Doebelin, Ernest O.: Measurement Systems: Application and Design. McGraw-Hill Book Co., Inc., 1966.
3. Iberall, Arthur S.: Attenuation of Oscillatory Pressures in Instrument Lines. J. Res. National Bureau of Standards, vol. 45, no. 1, July 1950, pp. 85-108.
4. Goldschmied, F. R.: On the Frequency Response of Viscous Compressible Fluids as a Function of the Stokes Number. J. Basic Eng., vol. 92, no. 2, June 1970, pp. 333-347.
5. Bergh, H.; and Tijdeman, H.: Theoretical and Experimental Results for the Dynamic Response of Pressure Measuring Systems. Rep. NLR-TR-F.238, National Aero- and Astronautical Research Inst., Amsterdam, Jan. 1965.
6. Nyland, Ted W.; Englund, David R.; and Anderson, Robert C.: On the Dynamics of Short Pressure Probes: Some Design Factors Affecting Frequency Response. NASA TN D-6151, 1971.
7. Hilsenrath, Joseph; Beckett, Charles W.; Benedict, William S.; Fano, Lilla; Hoge, Harold J.; Masi, Joseph F.; Nuttall, Ralph L.; Touloukian, Yerman S.; and Woolley, Harold W.: Tables of Thermal Properties of Gases. Circ. 564, National Bureau of Standards, Nov. 1, 1955.

NASA-Langley, 1971 —— 14 E-6377

NATIONAL AERONAUTICS AND SPACE ADMISTRATION
WASHINGTON, D.C. 20546

OFFICIAL BUSINESS
PENALTY FOR PRIVATE USE $300

FIRST CLASS MAIL

POSTAGE AND FEES PAID
NATIONAL AERONAUTICS AND
SPACE ADMINISTRATION

U.S.MAIL

018 001 C1 U 14 711001 S00903DS
DEPT OF THE AIR FORCE
AF SYSTEMS COMMAND
AF WEAPONS LAB (WLOL)
ATTN: E LOU BOWMAN, CHIEF TECH LIBRARY
KIRTLAND AFB NM 87117

POSTMASTER: If Undeliverable (Section 158 Postal Manual) Do Not Return

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered an important contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION OFFICE

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546